Wolfram Koepf

Computer Algebra Methods for Orthogonal Polynomials

Maple Worksheet

```
> restart;
```

Conversion of Recurrence and Difference Equations

```
> with(LREtools);
```

[*AnalyticityConditions*, *HypergeometricTerm*, *IsDesingularizable*, *REcontent*, *REcreate*, *REplot*, *REprimpart*, *REreduceorder*, *REtoDE*, *REtodelta*, *REtoproc*, *ValuesAtPoint*, *autodispersion*, *constcoeffsol*, *dAlembertiansols*, δ, *dispersion*, *divconq*, *firstlin*, *hypergeomsols*, *polysols*, *ratpolysols*, *riccati*, *shift*]

```
> RE:=n*f(n+2)-(n-1)*(n+1)*f(n+1)+f(n);
```

$$RE := n\, f(n+2) - (n-1)(n+1)\, f(n+1) + f(n)$$

Conversion to a difference equation:

```
> deltaexpr:=REtodelta(RE,f(n),{});
```

$$deltaexpr := n\, {LREtools_{\Delta}}_{n}^{2} + (-n^2 + 1 + 2n)\, {LREtools_{\Delta}}_{n} + n + 2 - n^2$$

```
> subs(LREtools[Delta][n]=Delta,deltaexpr);
```

$$n\, \Delta^2 + (-n^2 + 1 + 2n)\, \Delta + n + 2 - n^2$$

Now we convert back

```
> read "deltatore.mpl":
> deltatoRE(deltaexpr,f(n));
```

$$n\, f(n+2) - (n-1)(n+1)\, f(n+1) + f(n)$$

and compare with the original equation:

```
> RE;
```

$$n\, f(n+2) - (n-1)(n+1)\, f(n+1) + f(n)$$

```
>
```

Coefficients of Solution of Differential Equation

We define the polynomials σ and τ with arbitrary coefficients a,b,c,d,e:

```
> sigma:=a*x^2+b*x+c;
  tau:=d*x+e;
```

$$\sigma := a\, x^2 + b\, x + c$$

$$\tau := d\, x + e$$

and consider the differential equation

```
> DE:=sigma*diff(F(x),x$2)+tau*diff(F(x),x)-n*(a*n+d-a)*F(x);
```

$$DE := (a\, x^2 + b\, x + c)\left(\frac{d^2}{dx^2} F(x)\right) + (d\, x + e)\left(\frac{d}{dx} F(x)\right) - n\,(a\, n + d - a)\, F(x)$$

To convert the differential equation to a recurrence equation for the series coefficients, we load the gfun package.

```
> with(gfun):
> RE:=`diffeqtorec`(DE,F(x),A(j));
```

$$RE := (a\,j^2 + (d-a)\,j - n^2\,a - n\,d + a\,n)\,\mathrm{A}(j) + (b\,j^2 + (e+b)\,j + e)\,\mathrm{A}(j+1) + (c\,j^2 + 3\,c\,j + 2\,c)\,\mathrm{A}(j+2)$$

```
> map(factor,RE);
```

$$-(-j+n)\,(a\,n + d - a + a\,j)\,\mathrm{A}(j) + (j+1)\,(b\,j+e)\,\mathrm{A}(j+1) + c\,(j+1)\,(j+2)\,\mathrm{A}(j+2)$$

```
>
```

Computing the Recurrence Coefficients

Continuous case: We consider the three highest coefficients of the orthogonal polynomial:

```
> p:=k[n]*x^n+kprime[n]*x^(n-1)+kprimeprime[n]*x^(n-2);
```

$$p := k_n\,x^n + kprime_n\,x^{(n-1)} + kprimeprime_n\,x^{(n-2)}$$

The polynomial satisfies the differential equation DE=0 with:

```
> DE:=sigma*diff(p,x$2)+tau*diff(p,x)+lambda[n]*p;
```

$$DE := (a\,x^2 + b\,x + c)\left(\frac{k_n\,x^n\,n^2}{x^2} - \frac{k_n\,x^n\,n}{x^2} + \frac{kprime_n\,x^{(n-1)}\,(n-1)^2}{x^2} - \frac{kprime_n\,x^{(n-1)}\,(n-1)}{x^2} + \frac{kprimeprime_n\,x^{(n-2)}\,(n-2)^2}{x^2} - \frac{kprimeprime_n\,x^{(n-2)}\,(n-2)}{x^2}\right) + (d\,x+e)\left(\frac{k_n\,x^n\,n}{x} + \frac{kprime_n\,x^{(n-1)}\,(n-1)}{x} + \frac{kprimeprime_n\,x^{(n-2)}\,(n-2)}{x}\right) + \lambda_n\,(k_n\,x^n + kprime_n\,x^{(n-1)} + kprimeprime_n\,x^{(n-2)})$$

We collect coefficients:

```
> de:=collect(simplify(DE/x^(n-4)),x);
```

$$de := (-a\,k_n\,n + \lambda_n\,k_n + d\,k_n\,n + a\,k_n\,n^2)\,x^4 + (-3\,a\,kprime_n\,n + b\,k_n\,n^2 + a\,kprime_n\,n^2 + 2\,a\,kprime_n + \lambda_n\,kprime_n - d\,kprime_n + e\,k_n\,n - b\,k_n\,n + d\,kprime_n\,n)\,x^3 + (-5\,a\,kprimeprime_n\,n - 2\,d\,kprimeprime_n - e\,kprime_n - c\,k_n\,n - 3\,b\,kprime_n\,n + 2\,b\,kprime_n + c\,k_n\,n^2 + e\,kprime_n\,n + 6\,a\,kprimeprime_n + d\,kprimeprime_n\,n + \lambda_n\,kprimeprime_n + a\,kprimeprime_n\,n^2 + b\,kprime_n\,n^2)\,x^2 + (c\,kprime_n\,n^2 - 5\,b\,kprimeprime_n\,n + 2\,c\,kprime_n - 3\,c\,kprime_n\,n + e\,kprimeprime_n\,n + b\,kprimeprime_n\,n^2 + 6\,b\,kprimeprime_n - 2\,e\,kprimeprime_n)\,x - 5\,c\,kprimeprime_n\,n + 6\,c\,kprimeprime_n + c\,kprimeprime_n\,n^2$$

Equating the highest coefficient gives the already mentioned identity for λ:

```
> rule1:=lambda[n]=solve(coeff(de,x,4),lambda[n]);
```

$$rule1 := \lambda_n = -n\,(a\,n + d - a)$$

This can be substituted:

```
> de:=expand(subs(rule1,de));
```

$$de := 2\,x^3\,a\,kprime_n + 6\,x^2\,a\,kprimeprime_n + 2\,x^2\,b\,kprime_n + 6\,x\,b\,kprimeprime_n + 2\,x\,c\,kprime_n + 6\,c\,kprimeprime_n - x^3\,d\,kprime_n - 2\,x^2\,d\,kprimeprime_n - x^2\,e\,kprime_n - 2\,x\,e\,kprimeprime_n + x^2\,c\,k_n\,n^2 - x^2\,c\,k_n\,n - 2\,x^3\,a\,kprime_n\,n - 4\,x^2\,a\,kprimeprime_n\,n + x^3\,b\,k_n\,n^2 - x^3\,b\,k_n\,n + x^2\,b\,kprime_n\,n^2 - 3\,x^2\,b\,kprime_n\,n + x\,b\,kprimeprime_n\,n^2 - 5\,x\,b\,kprimeprime_n\,n + x\,c\,kprime_n\,n^2 - 3\,x\,c\,kprime_n\,n + c\,kprimeprime_n\,n^2 - 5\,c\,kprimeprime_n\,n + x^3\,e\,k_n\,n + x^2\,e\,kprime_n\,n + x\,e\,kprimeprime_n\,n$$

Equating the second highest coefficient gives k'[n] as rational multiple of k[n]:

```
> rule2:=kprime[n]=solve(coeff(de,x,3),kprime[n]);
```

$$rule2 := kprime_n = \frac{k_n\,n\,(e + b\,n - b)}{-2\,a + d + 2\,a\,n}$$

Equating the third highest coefficient gives k''[n] as rational multiple of k[n]:

```
> rule3:=kprimeprime[n]=solve(coeff(subs(rule2,de),x,2),kprimeprime[n]);
```

$$rule3 := kprimeprime_n = \frac{1}{2}\,k_n\,n\,(3\,b\,e + 5\,b^2\,n - 2\,b^2 + e^2\,n + 2\,e\,n^2\,b - 5\,e\,n\,b - e^2 - 4\,c\,n\,a + c\,n\,d + 2\,c\,n^2\,a + 2\,c\,a - c\,d + b^2\,n^3 - 4\,b^2\,n^2)\,/\,((-2\,a + d + 2\,a\,n)\,(-3\,a + d + 2\,a\,n))$$

Without loss of generality we consider the monic case, hence

```
> k[n]:=1;
```

$$k_n := 1$$

and therefore

```
> rule2;
```

$$kprime_n = \frac{n\,(e + b\,n - b)}{-2\,a + d + 2\,a\,n}$$

```
> rule3;
```

$$kprimeprime_n = n\,(3\,b\,e + 5\,b^2\,n - 2\,b^2 + e^2\,n + 2\,e\,n^2\,b - 5\,e\,n\,b - e^2 - 4\,c\,n\,a + c\,n\,d + 2\,c\,n^2\,a + 2\,c\,a - c\,d + b^2\,n^3 - 4\,b^2\,n^2)\,/\,(2\,(-2\,a + d + 2\,a\,n)\,(-3\,a + d + 2\,a\,n))$$

We would like to compute the coefficients a(n), b(n) and c(n) in the recurrence equation RE=0:

```
> RE:=x*P(n)-(a[n]*P(n+1)+b[n]*P(n)+c[n]*P(n-1));
```

$$RE := x\,\mathrm{P}(n) - a_n\,\mathrm{P}(n+1) - b_n\,\mathrm{P}(n) - c_n\,\mathrm{P}(n-1)$$

```
> RE:=subs({P(n)=p,P(n+1)=subs(n=n+1,p),P(n-1)=subs(n=n-1,p)},RE);
```

$$RE := x\,(x^n + kprime_n\,x^{(n-1)} + kprimeprime_n\,x^{(n-2)}) - a_n\,(x^{(n+1)} + kprime_{n+1}\,x^n + kprimeprime_{n+1}\,x^{(n-1)})$$

$$-b_n\,(x^n + kprime_n\,x^{(n-1)} + kprimeprime_n\,x^{(n-2)})$$
$$-c_n\,(x^{(n-1)} + kprime_{n-1}\,x^{(n-2)} + kprimeprime_{n-1}\,x^{(n-3)})$$

We substitute the already known formulas:

```
> RE:=subs({rule2,subs(n=n+1,rule2),subs(n=n-1,rule2),rule3,subs(n=n+1,rule3),subs(n=n-1,rule3)},RE);
```

$$RE := x\left(x^n + \frac{n\,(e+b\,n-b)\,x^{(n-1)}}{-2\,a+d+2\,a\,n} + n\,(3\,b\,e+5\,b^2\,n-2\,b^2+e^2\,n+2\,e\,n^2\,b-5\,e\,n\,b - e^2-4\,c\,n\,a+c\,n\,d+2\,c\,n^2\,a+2\,c\,a-c\,d+b^2\,n^3-4\,b^2\,n^2)\,x^{(n-2)}\,/\,(2\,(-2\,a+d+2\,a\,n)\,(-3\,a+d+2\,a\,n))\right) - a_n\left(x^{(n+1)} + \frac{(n+1)\,(e+b\,(n+1)-b)\,x^n}{-2\,a+d+2\,a\,(n+1)} + (n+1)\,(3\,b\,e+5\,b^2\,(n+1)-2\,b^2+e^2\,(n+1)+2\,e\,(n+1)^2\,b-5\,e\,(n+1)\,b-e^2 -4\,c\,(n+1)\,a+c\,(n+1)\,d+2\,c\,(n+1)^2\,a+2\,c\,a-c\,d+b^2\,(n+1)^3-4\,b^2\,(n+1)^2)\,x^{(n-1)}\,/\,(2\,(-2\,a+d+2\,a\,(n+1))\,(-3\,a+d+2\,a\,(n+1)))\right) - b_n\left(x^n + \frac{n\,(e+b\,n-b)\,x^{(n-1)}}{-2\,a+d+2\,a\,n} + n\,(3\,b\,e+5\,b^2\,n-2\,b^2+e^2\,n+2\,e\,n^2\,b-5\,e\,n\,b-e^2-4\,c\,n\,a + c\,n\,d+2\,c\,n^2\,a+2\,c\,a-c\,d+b^2\,n^3-4\,b^2\,n^2)\,x^{(n-2)}\,/\,(2\,(-2\,a+d+2\,a\,n)\,(-3\,a+d+2\,a\,n))\right) - c_n\left(x^{(n-1)} + \frac{(n-1)\,(e+b\,(n-1)-b)\,x^{(n-2)}}{-2\,a+d+2\,a\,(n-1)} + (n-1)\,(3\,b\,e +5\,b^2\,(n-1)-2\,b^2+e^2\,(n-1)+2\,e\,(n-1)^2\,b-5\,e\,(n-1)\,b-e^2-4\,c\,(n-1)\,a +c\,(n-1)\,d+2\,c\,(n-1)^2\,a+2\,c\,a-c\,d+b^2\,(n-1)^3-4\,b^2\,(n-1)^2)\,x^{(n-3)}\,/\,(2\,(-2\,a+d+2\,a\,(n-1))\,(-3\,a+d+2\,a\,(n-1)))\right)$$

```
> re:=simplify(numer(normal(RE))/x^(n-3)):
```

Equating the highest coefficient gives for monic polynomials

```
> rule4:=a[n]=solve(coeff(re,x,4),a[n]);
```

$$rule4 := a_n = 1$$

and equating the second highest coefficient yields

```
> rule5:=b[n]=factor(solve(subs(rule4,coeff(re,x,3)),b[n]));
```

$$rule5 := b_n = \frac{-2\,b\,n^2\,a+2\,b\,n\,a+2\,e\,a-2\,b\,n\,d-e\,d}{(d+2\,a\,n)\,(-2\,a+d+2\,a\,n)}$$

Finally equating the third highest coefficient yields

```
> rule6:=c[n]=factor(solve(subs(rule5,subs(rule4,coeff(re,x,2))),c[n]));
```

$$rule6 := c_n = -n\,(a\,n+d-2\,a)\,(4\,a^2\,n^2\,c-8\,a^2\,c\,n+4\,a^2\,c-a\,b^2\,n^2+4\,a\,c\,n\,d +2\,a\,b^2\,n+a\,e^2-a\,b^2-4\,a\,c\,d-b^2\,d\,n-b\,e\,d+c\,d^2+b^2\,d)\Big/((d-a+2\,a\,n)$$

$(-3\,a+d+2\,a\,n)\,(-2\,a+d+2\,a\,n)^2)$

```
>
```

Zeilberger's Algorithm

We load the package "hsum.mpl" from my book
"Hypergeometric Summation", Vieweg, Braunschweig/Wiesbaden, 1998

```
> read "hsum9.mpl";
```

Package "Hypergeometric Summation", Maple V - Maple 9

Copyright 1998-2004, Wolfram Koepf, University of Kassel

We define the hypergeometric summand of the Laguerre polynomials.

```
> laguerreterm:=(-1)^k/k!*binomial(n+alpha,n-k)*x^k;
```

$$laguerreterm := \frac{(-1)^k\,\mathrm{binomial}(n+\alpha,\,n-k)\,x^k}{k!}$$

and use Zeilberger's algorithm to detect a recurrence equation for the sum, hence for the Laguerre polynomials.

```
> LaguerreRE:=sumrecursion(laguerreterm,k,L(n));
```

$$LaguerreRE := (n+\alpha+1)\,\mathrm{L}(n)+(x-2\,n-\alpha-3)\,\mathrm{L}(n+1)+(n+2)\,\mathrm{L}(n+2)=0$$

Next, we detect the differential equation of the Laguerre polynomials from their hypergeometric representation.

```
> LaguerreDE:=sumdiffeq(laguerreterm,k,L(x));
```

$$LaguerreDE := x\left(\frac{d^2}{dx^2}\mathrm{L}(x)\right)-(x-\alpha-1)\left(\frac{d}{dx}\mathrm{L}(x)\right)+\mathrm{L}(x)\,n=0$$

Similarly, a recurrence equation w.r.t. α is obtained

```
> sumrecursion(laguerreterm,k,L(alpha));
```

$$(n+\alpha+1)\,\mathrm{L}(\alpha)-(x+\alpha+1)\,\mathrm{L}(\alpha+1)+x\,\mathrm{L}(\alpha+2)=0$$

The following computes the recurrence equation valid for the square of the Laguerre polynomials

```
> `rec*rec`(LaguerreRE,LaguerreRE,L(n));
```

$$\{(10-2\,x+29\,n+30\,n^2+25\,\alpha^2-5\,x\,\alpha+2\,n^4+13\,n^3+27\,\alpha+9\,\alpha^3+\alpha^4-x\,n^3-4\,x\,n^2$$
$$+7\,n^3\,\alpha+9\,n^2\,\alpha^2+35\,n^2\,\alpha+31\,n\,\alpha^2+5\,n\,\alpha^3-x\,\alpha^3-4\,x\,\alpha^2-3\,x\,n^2\,\alpha-3\,x\,n\,\alpha^2$$
$$-8\,x\,n\,\alpha-5\,x\,n+55\,n\,\alpha)\,\mathrm{L}(n)+(-66+70\,x-149\,n-124\,n^2-47\,\alpha^2+75\,x\,\alpha-6\,n^4$$
$$-45\,n^3-22\,x^2+2\,x^3-91\,\alpha-11\,\alpha^3-\alpha^4+x^3\,n-6\,x^2\,n^2-23\,x^2\,n+11\,x\,n^3+62\,x\,n^2$$
$$-15\,n^3\,\alpha-14\,n^2\,\alpha^2-84\,n^2\,\alpha-52\,n\,\alpha^2-6\,n\,\alpha^3+x^3\,\alpha-3\,x^2\,\alpha^2-17\,x^2\,\alpha+3\,x\,\alpha^3$$
$$+26\,x\,\alpha^2-9\,x^2\,n\,\alpha+22\,x\,n^2\,\alpha+14\,x\,n\,\alpha^2+82\,x\,n\,\alpha+115\,x\,n-153\,n\,\alpha)\,\mathrm{L}(n+1)+($$
$$110-102\,x+219\,n+160\,n^2+22\,\alpha^2-48\,x\,\alpha+6\,n^4+51\,n^3+26\,x^2-2\,x^3+82\,\alpha+2\,\alpha^3$$
$$-x^3\,n+6\,x^2\,n^2+25\,x^2\,n-11\,x\,n^3-70\,x\,n^2+9\,n^3\,\alpha+5\,n^2\,\alpha^2+57\,n^2\,\alpha+21\,n\,\alpha^2+n\,\alpha^3$$
$$+6\,x^2\,\alpha-6\,x\,\alpha^2+3\,x^2\,n\,\alpha-11\,x\,n^2\,\alpha-3\,x\,n\,\alpha^2-46\,x\,n\,\alpha-147\,x\,n+119\,n\,\alpha)$$
$$\mathrm{L}(n+2)+(x\,n^3+8\,x\,n^2+21\,x\,n+18\,x-2\,n^4-19\,n^3-66\,n^2-99\,n-n^3\,\alpha-8\,n^2\,\alpha$$

$$-21\,n\,\alpha-18\,\alpha-54)\,\mathrm{L}(n+3),\ \mathrm{L}(2)=\frac{1}{4}\_C_0\,\_C_2+\frac{9}{4}\_C_1\,\_C_3-\frac{3}{4}\_C_0\,\_C_3-\frac{3}{4}\_C_1\,\_C_2$$

$$+\frac{1}{4}\_C_1\,x\,\_C_2\,\alpha-\frac{1}{2}\_C_1\,\alpha\,\_C_3\,x+\frac{1}{4}\_C_0\,\alpha\,\_C_3\,x+\frac{1}{4}\_C_0\,\alpha^2\,\_C_2-\frac{1}{4}\_C_0\,\alpha^2\,\_C_3$$

$$-\frac{1}{4}\_C_1\,\alpha^2\,\_C_2+\frac{1}{4}\_C_1\,\alpha^2\,\_C_3+\frac{1}{4}\_C_1\,x^2\,\_C_3+\frac{1}{2}\_C_0\,\alpha\,\_C_2-\_C_0\,\alpha\,\_C_3+\frac{1}{4}\_C_0\,\_C_3\,x$$

$$-\_C_1\,\alpha\,\_C_2+\frac{3}{2}\_C_1\,\alpha\,\_C_3-\frac{3}{2}\_C_1\,\_C_3\,x+\frac{1}{4}\_C_1\,x\,\_C_2,\ \mathrm{L}(0)=\_C_0\,\_C_2,\ \mathrm{L}(1)=\_C_1\,\_C_3$$

}

the following is the differential equation for the square of the Laguerre polynomials

```
> `diffeq*diffeq`(LaguerreDE,LaguerreDE,L(x));
```

$$(-4\,x\,n+4\,n\,\alpha+2\,n)\,\mathrm{L}(x)+(4\,x\,n+2\,x^2+3\,\alpha+1+2\,\alpha^2-4\,x\,\alpha-4\,x)\left(\frac{d}{dx}\mathrm{L}(x)\right)$$

$$+(-3\,x^2+3\,x\,\alpha+3\,x)\left(\frac{d^2}{dx^2}\mathrm{L}(x)\right)+\left(\frac{d^3}{dx^3}\mathrm{L}(x)\right)x^2$$

and this is finally the differential equation for the product L(n,alpha,x)*L(m,beta,x) of the Laguerre polynomials.

```
> `diffeq*diffeq`(LaguerreDE,subs(alpha=beta,n=m,LaguerreDE),L(
  x));
```

$$(5\,\beta^2\,n^2\,x+4\,x\,n^2\,\alpha\,\beta+2\,\beta\,n^2\,x-2\,m^2\,\alpha\,x-m\,\beta\,\alpha^2-2\,m^2\,\beta\,x+m\,\alpha^2\,\beta^2+2\,m\,\alpha\,\beta^2$$

$$+\alpha\,n\,\beta^2-2\,\beta\,n\,\alpha^2-12\,m\,x^2\,n^2-5\,m^2\,\alpha^2\,x+m^2\,\beta^2\,x+12\,m^2\,x^2\,n-12\,m\,x^2\,n\,\alpha$$

$$+6\,m\,x\,n\,\alpha^2+12\,m\,\beta\,x^2\,n-6\,m\,\beta^2\,x\,n-4\,m^2\,x\,\alpha\,\beta-\alpha^3\,\beta\,n-\alpha^3\,m\,\beta+4\,x^3\,n\,\alpha$$

$$-4\,x^3\,n\,\beta+4\,x^3\,m\,\alpha-4\,x^3\,m\,\beta-4\,x^2\,\alpha^2\,n+x\,\alpha^3\,n-2\,x^2\,\alpha\,n^2+4\,x^2\,m\,\beta^2-4\,x^2\,\alpha\,n\,\beta$$

$$-x\,\alpha\,n\,\beta^2+4\,x^2\,m\,\beta\,\alpha+x\,m\,\beta\,\alpha^2+5\,x\,\beta\,n\,\alpha^2-5\,x\,m\,\alpha\,\beta^2+8\,x^3\,n^2-8\,x^3\,m^2-x\,m\,\beta^3$$

$$+2\,x^2\,m^2\,\beta+8\,x^2\,\beta^2\,n-5\,x\,\beta^3\,n-14\,x^2\,\beta\,n^2-8\,x^2\,m\,\alpha^2+5\,x\,m\,\alpha^3+14\,x^2\,m^2\,\alpha$$

$$+12\,m^2\,x^2-12\,x^2\,n^2-n\,\alpha^2-n\,\alpha^3-6\,x^2\,n\,\alpha+2\,x\,n^2\,\alpha+6\,x\,n\,\alpha^2-6\,x^2\,\alpha\,m+6\,\beta\,x^2\,m$$

$$+6\,\beta\,x^2\,n+6\,\alpha^2\,m\,x-6\,\beta^2\,m\,x-6\,\beta^2\,x\,n+\beta^3\,\alpha\,n+\beta^3\,\alpha\,m-\alpha^2\,n\,\beta^2-\alpha^2\,n^2\,x+\beta^4\,n$$

$$+4\,x^2\,n^3+n\,\beta^2-m\,\alpha^2+m\,\beta^2+2\,\beta^3\,n-4\,m^3\,x^2-m\,\alpha^4-2\,m\,\alpha^3+m\,\beta^3)\,\mathrm{L}(x)+(-\alpha^2$$

$$+\beta^2-2\,m\,x\,\beta+2\,x\,n\,\beta-9\,\beta\,x^2\,\alpha^2-6\,\beta^3\,x\,\alpha-2\,x\,m\,\alpha+4\,x^3\,n\,\alpha+20\,x^3\,n\,\beta$$

$$-20\,x^3\,m\,\alpha-4\,x^3\,m\,\beta+4\,x^2\,\alpha^2\,n-2\,x\,\alpha^3\,n+8\,x^2\,\alpha\,n^2-4\,x^2\,m\,\beta^2-6\,\beta\,\alpha\,m\,x$$

$$-16\,x^2\,\alpha\,n\,\beta+6\,x\,\alpha\,n\,\beta^2+16\,x^2\,m\,\beta\,\alpha-6\,x\,m\,\beta\,\alpha^2+2\,x\,\beta\,n\,\alpha^2-2\,x\,m\,\alpha\,\beta^2+8\,x^3\,\alpha^2$$

$$-3\,\beta\,\alpha^3+6\,\beta\,\alpha\,x\,n+3\,\beta^3\,\alpha-16\,x^3\,n^2+16\,x^3\,m^2-32\,m\,x^3+8\,m\,x^4+2\,x\,m\,\beta^3$$

$$-8\,x^2\,m^2\,\beta-12\,x^2\,\beta^2\,n+2\,x\,\beta^3\,n+8\,x^2\,\beta\,n^2+12\,x^2\,m\,\alpha^2-2\,x\,m\,\alpha^3-8\,x^2\,m^2\,\alpha$$

$$+21\,\beta^2\,x^2-9\,\beta^3\,x+\beta^4-12\,m^2\,x^2-2\,\alpha^3-\alpha^4+9\,x^2\,\alpha\,\beta^2+6\,x\,\alpha^3\,\beta+32\,x^3\,n+12\,x^2\,n^2$$

$$-20\,x^2\,n+16\,x^3\,\alpha-21\,x^2\,\alpha^2-10\,x^2\,\alpha+9\,x\,\alpha^3+10\,x\,\alpha^2-16\,x^2\,n\,\alpha-2\,x\,n\,\alpha^2+2\,x\,n\,\alpha$$

$$+10\,\beta\,x^2-10\,\beta^2\,x-2\,\alpha^2\,\beta+20\,m\,x^2+2\,\beta^3-5\,x^2\,\alpha^3-8\,\beta^2\,x^3+5\,\beta^3\,x^2-9\,x\,\alpha\,\beta^2$$

$$+28x^2\alpha m+9\beta x\alpha^2+16\beta x^2 m-28\beta x^2 n-8\alpha^2 m x+2\beta^2 m x+8\beta^2 x n-\alpha^4\beta$$
$$-\alpha^3\beta^2-8x^4 n-4x^4\alpha+4\beta x^4-\beta^4 x+\beta^3\alpha^2+\beta^4\alpha-16\beta x^3+2\alpha\beta^2+\alpha^4 x)$$
$$\left(\frac{d}{dx}\mathrm{L}(x)\right)+(9\beta x^2\alpha^2+3\beta^3 x\alpha-16x^3 n\alpha-24x^3 n\beta+24x^3 m\alpha+16x^3 m\beta$$
$$+2x^2\alpha^2 n-2x^2 m\beta^2+12x^2\alpha n\beta-12x^2 m\beta\alpha-15x^3\alpha^2+8x^3 n^2-8x^3 m^2+52 m x^3$$
$$-20 m x^4+6x^2\beta^2 n-6x^2 m\alpha^2-25\beta^2 x^2+6\beta^3 x-9x^2\alpha\beta^2-3x\alpha^3\beta-52x^3 n$$
$$+16x^2 n-26x^3\alpha+25x^2\alpha^2+8x^2\alpha-6x\alpha^3-7x\alpha^2+18x^2 n\alpha-8\beta x^2+7\beta^2 x$$
$$-16 m x^2+7x^2\alpha^3+15\beta^2 x^3-7\beta^3 x^2+6x\alpha\beta^2-18x^2\alpha m-6\beta x\alpha^2-18\beta x^2 m$$
$$+18\beta x^2 n+20x^4 n+10x^4\alpha-10\beta x^4+\beta^4 x+26\beta x^3-\alpha^4 x)\left(\frac{d^2}{dx^2}\mathrm{L}(x)\right)+(-8\beta^2 x^3$$
$$+6\beta^2 x^2+8x^3\alpha^2+20x^3 n+16 m x^4-16x^4 n-20 m x^3+10x^3\alpha-10\beta x^3-6x^2\alpha^2$$
$$-8x^4\alpha+8\beta x^4+2x^2\alpha\beta^2-8x^3 m\alpha+8x^3 n\alpha-2\beta x^2\alpha^2-8x^3 m\beta+8x^3 n\beta$$
$$-2x^2\alpha^3+2\beta^3 x^2)\left(\frac{d^3}{dx^3}\mathrm{L}(x)\right)$$
$$+(2x^4\alpha-2\beta x^4-x^3\alpha^2+\beta^2 x^3-4 m x^4+4x^4 n)\left(\frac{d^4}{dx^4}\mathrm{L}(x)\right)$$

\>

Petkovsek-van Hoeij Algorithm

We would like to find a simple representation of for

```
> s:=Sum(binomial(n-2*k,k)*(-4/27)^k,k=0..floor(n/3));
```

$$s:=\sum_{k=0}^{\mathrm{floor}\left(\frac{n}{3}\right)}\mathrm{binomial}(n-2k,k)\left(\frac{-4}{27}\right)^k$$

```
> summand:=binomial(n-2*k,k)*(-4/27)^k;
```

$$summand:=\mathrm{binomial}(n-2k,k)\left(\frac{-4}{27}\right)^k$$

```
> RE:=sumrecursion(summand,k,S(n));
```

$$RE:=2(n+3)\,\mathrm{S}(n)+3(n+4)\,\mathrm{S}(n+1)-9(n+2)\,\mathrm{S}(n+2)=0$$

```
> res:=`LREtools/hsols`(RE,S(n));
```

$$res:=\left[\left(\frac{-1}{3}\right)^n,\left(\frac{2}{3}\right)^n\left(\frac{4}{3}+n\right)\right]$$

```
> result:=alpha*op(1,res)+beta*op(2,res);
```

$$result:=\alpha\left(\frac{-1}{3}\right)^n+\beta\left(\frac{2}{3}\right)^n\left(\frac{4}{3}+n\right)$$

```
> sol:=solve({
eval(subs(Sum=sum,s=result),n=0),
eval(subs(Sum=sum,s=result),n=1)},
```

```
{alpha,beta});
```

$$sol := \{\beta = \frac{2}{3}, \alpha = \frac{1}{9}\}$$

```
> result:=subs(sol,result);
```

$$result := \frac{\left(\frac{-1}{3}\right)^n}{9} + \frac{2\left(\frac{2}{3}\right)^n\left(\frac{4}{3}+n\right)}{3}$$

```
> [seq(add(binomial(n-2*k,k)*(-4/27)^k,k=0..floor(n/3)),n=0..10
)]; n:='n':
```

$$\left[1, 1, 1, \frac{23}{27}, \frac{19}{27}, \frac{5}{9}, \frac{313}{729}, \frac{79}{243}, \frac{59}{243}, \frac{3527}{19683}, \frac{2579}{19683}\right]$$

```
> [seq(result,n=0..10)]; n:='n':
```

$$\left[1, 1, 1, \frac{23}{27}, \frac{19}{27}, \frac{5}{9}, \frac{313}{729}, \frac{79}{243}, \frac{59}{243}, \frac{3527}{19683}, \frac{2579}{19683}\right]$$

```
>
```

Factorization of Recurrence Equations

```
> tau:='tau':
> read "FactorOrder4-discrete";
Warning, the name delta has been redefined
```

$$\_Env\_LRE\_x := x$$

$$\_Env\_LRE\_tau := \tau$$

```
> RE:=f(x+2)-(x+1)*f(x+1)+x^2*f(x);
```

$$RE := f(x+2) - (x+1)\, f(x+1) + x^2\, f(x)$$

```
> RE:=collect((x^2+x-1)*RE+x^3*subs(x=x+1,RE)+x*(x+1)*subs(x=x+
2,RE),f,factor);
```

$$RE := (x+1)(3x^2+6x-1)\, f(x+2) + (x+1)(x^4+x^3-x^2-x+1)\, f(x+1) + (x^2+x-1)\, x^2\, f(x) - x(4x+3)\, f(x+3) + x(x+1)\, f(x+4)$$

```
> RE:=subs({seq(f(x+k)=tau^k,k=0..4)},RE);
```

$$RE := (x+1)(3x^2+6x-1)\,\tau^2 + (x+1)(x^4+x^3-x^2-x+1)\,\tau + (x^2+x-1)\,x^2 - x(4x+3)\,\tau^3 + x(x+1)\,\tau^4$$

```
> fact:=FactorOrder4(RE);
```

$$fact := \{\tau^2 + (-x-1)\,\tau + x^2\}$$

```
>
```

Orthogonal Polynomial Solutions of Recurrence Equations

```
> read "retode.mpl";
```

Package "REtoDE", Maple V - Maple 8

Copyright 2000-2002, Wolfram Koepf, University of Kassel

Example recurrence

```
> RE:=P(n+2)-(x-n-1)*P(n+1)+alpha*(n+1)^2*P(n)=0;
```

$$RE := \mathrm{P}(n+2) - (x-n-1)\,\mathrm{P}(n+1) + \alpha\,(n+1)^2\,\mathrm{P}(n) = 0$$

Classical continuous solutions

```
> REtoDE(RE,P(n),x);
```

Warning: parameters have the values, $\{b = 2\,c,\ a = 0,\ \alpha = \frac{1}{4},\ c = c,\ d = -4\,c,\ e = 0\}$

$$\left[\frac{1}{2}(2\,x+1)\left(\frac{\partial^2}{\partial x^2}\mathrm{P}(n,x)\right) - 2\,x\left(\frac{\partial}{\partial x}\mathrm{P}(n,x)\right) + 2\,n\,\mathrm{P}(n,x) = 0,\right.$$
$$\left.\left[I = \left[\frac{-1}{2},\infty\right],\ \rho(x) = 2\,\mathbf{e}^{(-2\,x)}\right],\ \frac{k_{n+1}}{k_n} = 1\right]$$

Classical discrete solutions

```
> REtodiscreteDE(RE,P(n),x);
```

Warning: parameters have the values, $\{\alpha = \frac{f^2-1}{4\,f^2},\ f = f,\ d = d,$

$c = -\frac{1}{4}f^2\,d + \frac{1}{4}d + \frac{1}{2}g\,d\,f + \frac{1}{2}g\,d,\ a = 0,\ g = g,\ e = -g\,d,\ b = -\frac{1}{2}f\,d - \frac{1}{2}d\}$

$$\left[\frac{1}{2}\,\frac{(f+2\,f\,x-1)\,\Delta(\mathrm{Nabla}(\mathrm{P}(n,f\,x+g),x),x)}{f} - \frac{2\,x\,\Delta(\mathrm{P}(n,f\,x+g),x)}{f+1}\right.$$
$$+ \frac{2\,n\,\mathrm{P}(n,f\,x+g)}{(f+1)\,f} = 0,\ \left[\sigma(x) = \frac{f}{2} + x - \frac{1}{2} - g,\ \sigma(x) + \tau(x) = \frac{(f-1)\,(f+2\,x+1-2\,g)}{2\,(f+1)}\right],$$
$$\left.\rho(x) = \left(\frac{f-1}{f+1}\right)^x,\ \frac{k_{n+1}}{k_n} = \frac{1}{f}\right]$$

```
>
```